

UNCLASSIFIED

AR-006-736



ELECTRONICS RESEARCH LABORATORY

# Electronic Warfare Division

RESEARCH NOTE  
ERL-0559-RN

THE DESIGN OF TOPWALL WAVEGUIDE DIRECTIONAL COUPLERS

Kenneth M. Harvey

ABSTRACT (U)

This report outlines a procedure for the design of topwall waveguide directional couplers. It is accompanied by a listing of a GWBASIC programme WGCUP.BAS that renders the whole procedure quick, easy and accurate.

JUNE 1991

© Commonwealth of Australia 1991

COPY No.

APPROVED FOR PUBLIC RELEASE

Postal Address: Director, Electronics Research Laboratory,  
PO Box 1600, Salisbury, South Australia, 5108.

UNCLASSIFIED

ERL-0559-RN

*This work is Copyright. Apart from any use as permitted under the Copyright Act 1968, no part may be reproduced by any process without permission from the Australian Government Publishing Service. Requests and enquiries concerning reproduction and rights should be directed to the Manager, AGPS Press, GPO Box 84, Canberra ACT 2601.*

## CONTENTS

|   |   |   |
|---|---|---|
| 1 | INTRODUCTION .....  | 1 |
| 2 | THEORY.....   | 1 |
|   | 2.1 Basic operation.....                                      | 1 |
|   | 2.2 Bethe's aperture coupling theory.....                     | 1 |
|   | 2.3 Corrections for large aperture and finite thickness. .... | 2 |
| 3 | PROCEDURE FOR CALCULATING COUPLING COEFFICIENTS.....          | 2 |
| 4 | PROCEDURE FOR CALCULATING HOLE SIZES.....                     | 3 |
| 5 | MEASURED COUPLERS .....                                       | 3 |
|   | REFERENCES .....  | 5 |

### LIST OF TABLES

|   |  |    |
|---|--|----|
| 1 | Element values of Chebyshev arrays for $\cos \theta = c$ ..... | 10 |
|---|--|----|

### LIST OF FIGURES

|   |   |   |
|---|---|---|
| 1 | Waveguide coupler operation.....        | 6 |
| 2 | Coupler dimensions.....                 | 6 |
| 3 | S <sub>21</sub> of a 9 dB Coupler.....  | 7 |
| 4 | S <sub>21</sub> of a 18 dB Coupler..... | 7 |
| 5 | Drawings of a 9 dB Coupler .....        | 8 |

### LIST OF APPENDICES

|   |                      |    |
|---|----------------------|----|
| I | PROGRAM LISTING..... | 11 |
|---|----------------------|----|

THIS IS A BLANK PAGE

## 1 INTRODUCTION

The design of waveguide directional couplers has been covered extensively in the technical literature. However, the extraction of an easy-to-understand procedure suitable for computer aided design has not been quite so forthcoming. This paper, together with a computer programme WGCUP.BAS, coded in GWBASIC, renders the design procedure a much less formidable task. This work concentrates on the Chebyshev distribution as it gives a much higher directivity over a waveguide band than the binomial distribution. It also restricts the design to topwall couplers as this gives a flatter frequency response than the sidewall coupler, although sidewall couplers may be more suitable for high power applications. Only circular coupling holes are considered.

The theory outlined here is derived from Ref 1. This report is not intended as a treatise on the theory of couplers but rather as support documentation for the programme.

## 2 THEORY

### 2.1 Basic operation

Consider two waveguides coupled together by a series of holes in their common broad wall as shown in Figure 1. If the holes are spaced by  $\lambda_g/4$ , then wave A in waveguide 1 will couple into waveguide 2 by constructive interference to form wave B in the forward direction. Destructive interference will cause almost no wave to propagate in waveguide 2 in the reverse direction. The ratio of the forward to backward wave is the directivity of the coupler and it is usually desired to maximise this figure across the whole band. This is done by having a large number of coupling holes of varying diameter so that the coupling is tapered along the waveguide. The coupling for each hole is here designed to follow a Chebyshev taper as this gives a maximum directivity over a given band. It is first important to understand basic aperture coupling theory and then to extend it for electrically large holes in a finite thickness wall.

### 2.2 Bethe's aperture coupling theory (see Ref 2)

Assuming the waveguides are rectangular (see Figure 2) and propagate the dominant  $H_{10}$  mode, then the forward coupled wave from one hole is given by:-

$$U = \frac{j2\pi}{ab\lambda_g} \left[ M_x H_x^{(1)} H_x^{(2)} + M_z H_z^{(1)} H_z^{(2)} - P_y E_y^{(1)} E_y^{(2)} \right]$$

where  $M_x$  and  $M_z$  are the components of the magnetic polarizability of the aperture in the x and

z directions, and  $P_y$  is the electric polarizability.  $H_x^{(1)}$  is the magnitude of the x component of

the magnetic vector in the first guide at the aperture while  $H_x^{(2)}$  is the corresponding value for the second waveguide. The other definitions are similar. a,b and t are the dimensions shown in Figure 2 and  $\lambda_g$  is the waveguide wavelength. When the guides are identical (as assumed here)

then  $H_x^{(1)} = H_x^{(2)} = H_x$  etc.

The values of the field components in the +ve direction are given by:-

$$H_x = -\sin\left(\frac{\pi x}{a}\right) \cdot \exp(-j\gamma z)$$

$$H_z = j\frac{\lambda_g}{2a} \cos\left(\frac{\pi x}{a}\right) \cdot \exp(-j\gamma z)$$

$$E_y = \frac{\lambda_g}{\lambda} \sin\left(\frac{\pi x}{a}\right) \cdot \exp(-j\gamma z)$$

where  $x$  is the distance of the hole from the side wall and  $\lambda$  is the wavelength in free space.

For small apertures the polarizabilities  $M_x, M_z$  and  $P_y$  are given by:-

$$M_x = M_z = \frac{d^3}{6} \text{ and } P_y = \frac{d^3}{12}$$

where  $d$  is the diameter of the coupling hole.

Substituting for these expressions gives:-

$$U = j \frac{2\pi d^3}{ab\lambda_g} \left[ \frac{\sin^2\left(\frac{\pi x}{a}\right)}{6} + \frac{\lambda_g \cos^2\left(\frac{\pi x}{a}\right)}{24a^2} - \frac{\lambda_g \sin^2\left(\frac{\pi x}{a}\right)}{12\lambda^2} \right]$$

### 2.3 Corrections for large aperture and finite thickness.

Each of the terms in the above equation should be multiplied by the following expression to correct for the finite dimensions of the coupling hole:-

$$K = \frac{\tan\left(\frac{\pi\lambda_0}{2\lambda}\right)}{\left(\frac{\pi\lambda_0}{2\lambda}\right)} \cdot \exp\left[-\frac{2\pi A t}{\lambda_0} \sqrt{1 - \frac{\lambda_0^2}{\lambda^2}}\right]$$

where the first term takes account of the effects of the aperture resonant wavelength  $\lambda_0$ . The exponential term corrects for the finite thickness  $t$  of the coupling wall, where  $A$  is an empirical factor which takes into account an apparent extra electrical thickness. This takes the form  $At=ad+t$ .

For the magnetic coupling terms,  $At=1.0064t+0.04095d$  and for the electric coupling term,  $At=1.0103t+.02895d$  (See Ref 1, page 338).

Also, for round holes  $\lambda_0 = 1.705 d$  for the magnetic coupling terms and  $\lambda_0 = 1.305 d$  for the electric coupling term.

### 3 PROCEDURE FOR CALCULATING COUPLING COEFFICIENTS (SEE REF 3)

Programme WGCUP.BAS (see Appendix I) computes the coupling coefficients for each hole of an  $N$  hole coupler. The programme requests the lowest and highest frequencies and calculates their guide wavelengths and the bandwidth ratio  $P \cdot \cos \theta$  is calculated from these where  $\theta$  is the element spacing in electrical degrees at the lowest frequency. Table 1 lists the element values of Chebyshev arrays for  $\cos \theta = c$  for  $N=3$  to 8. This table forms part of the programme and is obtained from Ref 4. The coupling coefficients are then normalised to the sum of all coefficients and converted to decibels. Now the array so formed would pass 100% of the wave (ie a 0 dB coupler). For a coupling value of  $C$  dB, this figure must be subtracted from all the coupling values.

In many cases this procedure would give rise to higher coupling values per hole (hence, larger hole diameters) than could be realised because of overlapping. A common procedure is to superimpose a number of arrays in a staggered fashion. This effectively lengthens the coupling section and enables the hole sizes to be kept to manageable values without degrading the directivity. Let us consider a design example.

Design a 10 dB coupler for the frequency band 12 to 18 GHz.

The waveguide size is 0.622" x 0.311". Assume a 5 element array.

The programme gives the hole spacing as 0.2609" and the coupling coefficients as 1., 2.900, 3.951, 2.900 and 1.

Now add 2 further identical arrays in a staggered fashion to produce central hole sizes that are near equal:-

|      |      |      |      |      |      |      |      |      |      |      |
|------|------|------|------|------|------|------|------|------|------|------|
| 1.00 | 2.90 | 3.95 | 2.90 | 1.00 |      |      |      |      |      |      |
|      |      |      | 1.00 | 2.90 | 3.95 | 2.90 | 1.00 |      |      |      |
|      |      |      |      |      |      | 1.00 | 2.90 | 3.95 | 2.90 | 1.00 |

Adding up these coefficients to create one 11 hole array gives:-

1.00 2.90 3.95 3.90 3.90 3.95 3.90 3.90 3.95 2.90 1.00

Normalising these values against their sum and converting to dBs gives:-

-30.9 -21.7 -19.0 -19.1 -19.1 -19.0 -19.1 -19.1 -19.0 -21.7 -30.9

Not forgetting that this is a 10 dB coupler we now have:-

-40.9 -31.7 -29.0 -29.1 -29.1 -29.0 -29.1 -29.1 -29.0 -31.7 -40.9

Now this is the design for a single series of holes. It may be more convenient to arrange for two parallel rows of coupling holes, one on each side of the centreline of the waveguide as this keeps the hole sizes down. In this case, each array is then designed for 16 dB coupling, since doubling the number of coupling elements always increases the overall coupling by 6 dB. Thus, our new array has the following coupling values in dBs:-

-46.9 -37.7 -35.0 -35.1 -35.1 -35.0 -35.1 -35.1 -35.0 -37.7 -46.9

The programme asks for the number of superimposed arrays and whether there are to be one or two rows of holes.

#### 4 PROCEDURE FOR CALCULATING HOLE SIZES

The programme also computes the hole sizes from the coupling coefficients calculated above. It asks for the displacement of the holes from the side wall as a fraction of the waveguide broadwall dimension. A factor of 0.203 is common for 2 hole apertures. Since the programme is basically an analytic one, an iterative method is used to calculate the required hole size. The formula outlined above is used. Initial values are computed from the uncorrected Bethe's formula for  $x=0.25a$ .

The user should check hole sizes to ensure that there is no overlapping.

The programme WGCUP.BAS undertakes a full design starting with the requirement and finishing with an array of hole diameters. The whole design procedure takes seconds on an IBM PC.

#### 5 MEASURED COUPLERS

The programme was written as an aid to the fabrication of millimetre-wave couplers up to 110 GHz. However, it was decided to validate the programme at 26-40 GHz as that represented the upper frequency limit of the available network analyser.

Two couplers of 9 and 18 dB design were constructed using a split block construction technique. The coupling holes were drilled in a 0.8 mm thick intervening brass plate. For the 9 dB coupler a 24 doublet hole design was chosen. This comprised 5 superimposed 8 doublet hole arrays. Figure 5 shows the drawings of this coupler. The 18 dB coupler had 16 doublets comprising 3 superimposed 8 doublet hole arrays. The measured results are shown in Figures 3 and 4. The coupling at the design frequency is very close to the calculated value. Measuring the directivity was difficult as it calls for excellent transitions and loads. However, the directivities for both couplers were at least 35 dB.



## REFERENCES

- 1 "Improved Single and Multiaperture Waveguide Coupling Theory, Including Explanation of Mutual Interactions" R.Levy IEEE Trans MTT-28 No 4 April 1980 pp 331-338
- 2 "Microwave Coupling by Large Apertures" S.Cohn Proc IRE Vol 40 June 1952 pp 696-699
- 3 "The Design of Multi-hole Coupling Arrays" E.D.Hensperger Microwave Journal Aug 1959 pp 38-42
- 4 "Directional Couplers" R.Levy in Advances in Microwaves Vol 1 1966 pp 115-209




Figure 1 Waveguide coupler operation.




Figure 2 Coupler dimensions.




Figure 3 S21 of a 9 dB Coupler.





Figure 4 S21 of a 18 dB Coupler.

WAVEGUIDE COUPLER PART A  
MATERIAL: ALUMINIUM



WAVEGUIDE COUPLER PART B  
MATERIAL: ALUMINIUM






Figure 5 Drawings of a 9 dB Coupler .

WAVEGUIDE APERTURE PLATE - MATERIAL: 0.8mm BRASS



WAVEGUIDE APERTURE PLATE DETAIL



HOLE DIAMETERS

|        |        |        |        |
|--------|--------|--------|--------|
| A 1.31 | D 2.31 | G 2.29 | J 2.29 |
| B 1.78 | E 2.34 | H 2.34 | K 2.29 |
| C 2.13 | F 2.29 | I 2.34 | L 2.34 |

Figure 5 (continued).

Table 1 Element values of Chebyshev arrays for  $\cos \theta = c$ .

| $n$ | $a_1$ | $a_2$      | $a_3$                         | $a_4$   | $a_5$                             | $a_6$                         | $a_7$      | $a_8$ |
|-----|-------|------------|-------------------------------|---|-----------------------------------|-------------------------------|------------|-------|
| 3   | 1     | $2(1-c^2)$ | 1                             |   |                                   |                               |            |       |
| 4   | 1     | $3(1-c^2)$ | $3(1-c^2)$                    | 1   |                                   |                               |            |       |
| 5   | 1     | $4(1-c^2)$ | $6(1-c^2)(1-\frac{1}{3}c^2)$  | $4(1-c^2)$                                    | 1                                 |                               |            |       |
| 6   | 1     | $5(1-c^2)$ | $10(1-c^2)(1-\frac{1}{3}c^2)$ | $10(1-c^2)(1-\frac{1}{3}c^2)$                 | $5(1-c^2)$                        | 1                             |            |       |
| 7   | 1     | $6(1-c^2)$ | $15(1-c^2)(1-\frac{1}{3}c^2)$ | $20(1-c^2)(1-\frac{4}{3}c^2+\frac{1}{10}c^4)$ | $15(1-c^2)(1-\frac{1}{3}c^2)$     | $6(1-c^2)$                    | 1          |       |
| 8   | 1     | $7(1-c^2)$ | $21(1-c^2)(1-\frac{1}{3}c^2)$ | $35(1-c^2)(1-c^2+\frac{1}{3}c^4)$             | $35(1-c^2)(1-c^2+\frac{1}{3}c^4)$ | $21(1-c^2)(1-\frac{1}{3}c^2)$ | $7(1-c^2)$ | 1     |

## APPENDIX I

## PROGRAM LISTING

```

10 REM DESIGN OF A WAVEGUIDE DIRECTIONAL COUPLER
20 REM USES LEVY'S CORRECTION
30 REM SEE "IMPROVED SINGLE AND MULTIAPERTURE WAVEGUIDE COUPLING THEORY,
40 REM INCLUDING EXPLANATION OF MUTUAL INTERACTIONS" RALPH LEVY
50 REM IEEE TRANS MTT-28 NO 4 APRIL 1980 PP 331-338
60 REM ALSO "ANALYSIS AND SYNTHESIS OF WAVEGUIDE MULTIAPERTURE
70 REM DIRECTIONAL COUPLERS" IEEE TRANS MTT-16 NO 12 DEC 1968 PP 995-1006
80 REM X IS THE DISTANCE OF THE HOLES FROM THE SIDE WALL
90 DIM A(40),AA(40),AS(8,40),CO(40)
100 FOR J=1 TO 8 STEP 1
110 FOR K=1 TO 30 STEP 1
120 AS(J,K)=0
130 A(K)=0
140 NEXT K
150 NEXT J
160 PI=3.141592654#
170 INPUT "REQUIRED COUPLING (DB) ";CC
180 CC=ABS(CC)
190 INPUT "LOWEST FREQUENCY (GHZ) ";FL
200 INPUT "HIGHEST FREQUENCY (GHZ) ";FH
210 FL=FL*1E+09
220 FH=FH*1E+09
230 C=2.998E+08
240 LL=C/FL
250 LH=C/FH
260 F=(FL+FH)/2
270 FF=F/1E+09
280 PRINT "DESIGN FREQUENCY= ";FF;" GHZ"
290 L=C/F
300 INPUT "A OF WAVEGUIDE (INS) ";A
310 A=A*25.4/1000
320 INPUT "B OF WAVEGUIDE (INS) ";B
330 B=B*25.4/1000
340 INPUT "THICKNESS OF COMMON WALL (INS) ";T
350 T=T*25.4/1000
360 INPUT "X AS A FRACTION OF A ";X
370 LL=1/SQR(1/LL/LL-1/(2*A)^2)
380 LH=1/SQR(1/LH/LH-1/(2*A)^2)
390 LG=1/SQR(1/L/L-1/(2*A)^2)
400 P=LL/LH
410 PH=PI/(1+P)
420 C=COS(PH)*COS(PH)
430 CA=1-C
440 SP=LL*LH/(2*(LL+LH))
450 SP=SP*1000/25.4
460 PRINT "HOLE SPACING= ";SP;" INS"
470 INPUT "NO. OF HOLES (3-8) ";N
480 REM NOW SELECT THE APPROPRIATE CHEBYSHEV COEFFICIENTS
490 IF N=3 THEN GOTO 580
500 IF N=4 THEN GOTO 640
510 IF N=5 THEN GOTO 710
520 IF N=6 THEN GOTO 790
530 IF N=7 THEN GOTO 880
540 IF N=8 THEN GOTO 980
550 GOTO 470
560 REM*****
570 REM N=3
580 A(1)=1

```

```

590 A(2)=2*CA
600 A(3)=1
610 GOTO 1070
620 REM*****
630 REM N=4
640 A(1)=1
650 A(2)=3*CA
660 A(3)=A(2)
670 A(4)=1
680 GOTO 1070
690 REM*****
700 REM N=5
710 A(1)=1
720 A(2)=4*CA
730 A(3)=6*CA*(1-C/3)
740 A(4)=A(2)
750 A(5)=1
760 GOTO 1070
770 REM*****
780 REM N=6
790 A(1)=1
800 A(2)=5*CA
810 A(3)=10*CA*(1-C/2)
820 A(4)=A(3)
830 A(5)=A(2)
840 A(6)=A(1)
850 GOTO 1070
860 REM*****
870 REM N=7
880 A(1)=1
890 A(2)=6*CA
900 A(3)=15*CA*(1-.6*C)
910 A(4)=20*CA*(1-.8*C+.1*C*C)
920 A(5)=A(3)
930 A(6)=A(2)
940 A(7)=1
950 GOTO 1070
960 REM*****
970 REM N=8
980 A(1)=1
990 A(2)=7*CA
1000 A(3)=21*CA*(1-2*C/3)
1010 A(4)=35*CA*(1-C+.2*C*C)
1020 A(5)=A(4)
1030 A(6)=A(3)
1040 A(7)=A(2)
1050 A(8)=1
1060 REM NOW NORMALISE THE COUPLING COEFFICIENTS
1070 SUM=0
1080 FOR I=1 TO N STEP 1
1090 SUM=SUM+A(I)
1100 NEXT I
1110 FOR I=1 TO N STEP 1
1120 AA(I)=A(I)/SUM
1130 AA(I)=20*LOG(AA(I))/2.302585
1140 NEXT I
1150 REM *****
1160 INPUT "HOW MANY SUPERIMPOSED ARRAYS ";NA
1170 NR=1
1180 INPUT "TWO ROWS OF HOLES? (Y/N) ";IN$
1190 IF IN$="Y" THEN NR=2
1200 IF IN$="y" THEN NR=2
1210 N2=INT(N/2)
1220 NH=N+(NA-1)*N2

```



```

1230 PRINT "TOTAL NUMBER OF HOLES PER ROW= ";NH
1240 REM NOW GENERATE THE STAGGERED ARRAYS OF COUPLING COEFFICIENTS AS(J,K)
1250 FOR J=1 TO NA STEP 1
1260 FOR K=1+(J-1)*N2 TO NH STEP 1
1270 AS(J,K)=A(K-(J-1)*N2)
1280 NEXT K
1290 NEXT J
1300 REM ADD UP THE SUPERIMPOSED ARRAY COEFFICIENTS
1310 FOR K=1 TO NH STEP 1
1320 SUM=0
1330 FOR J=1 TO NA STEP 1
1340 SUM=SUM+AS(J,K)
1350 NEXT J
1360 CO(K)=SUM
1370 NEXT K
1380 REM NORMALIZE THE COUPLING COEFFICIENTS
1390 SUM=0
1400 FOR I=1 TO NH STEP 1
1410 SUM=SUM+CO(I)
1420 NEXT I
1430 REM CONVERT TO DECIBELS AND ADD IN THE COUPLER VALUE AS WELL AS
1440 REM ADDING 6 DECIBELS FOR A 2 ROW COUPLER (OR 0 DBS FOR A 1 ROW COUPLER)
1450 FOR I=1 TO NH STEP 1
1460 CO(I)=20*LOG(SUM/CO(I))/2.302585+CC+6.0206*(NR-1)
1470 DB=CO(I)
1480 GOSUB 1540
1490 PRINT "HOLE DIAMETER= ";DI;" INS  = ";DM;" MM FOR ";DB;" DB"
1500 NEXT I
1510 END
1520 REM *****
1530 REM THIS IS THE ITERATIVE ROUTINE FOR CALCULATING THE HOLE DIAMETERS
1540 REM START OFF WITH A GUESS FOR THE INITIAL VALUE OF D
1550 REM USE USE SIMPLIFIED FORMULA WITHOUT CORRECTIONS
1560 D1=(12*A*A*B/(PI*10^(DB/20)))^.333333
1570 D=D1
1580 GOSUB 1800
1590 C1=C
1600 REM HAVE ANOTHER GUESS FOR D (1 DB DIFFERENCE FROM THE INITIAL GUESS)
1610 D2=(12*A*A*B/(PI*10^((DB-1)/20)))^.333333
1620 D=D2
1630 GOSUB 1800
1640 C2=C
1650 REM PERFORM A LINEAR PREDICTION
1660 P=(C1-C2)/(D1-D2)
1670 Q=(C1*D2-D1*C2)/(D2-D1)
1680 DF=(DB-Q)/P
1690 D=DF
1700 GOSUB 1800
1710 C2=C
1720 D2=DF
1730 REM IS IT CLOSE ENOUGH TO THE COUPLING VALUE REQUIRED?
1740 IF ABS(DB-C)>.0001 THEN GOTO 1660
1750 DI=DF*1000/25.4
1760 DM=DF*1000
1770 RETURN
1780 END
1790 REM *****
1800 REM SUBROUTINE TO COMPUTE THE COUPLING THROUGH A HOLE
1810 REM IN THE BROAD WALL OF A WAVEGUIDE
1820 REM KE IS THE THICKNESS CORRECTION FOR THE ELECTRIC TERM
1830 AET=1.0103*T+.0579*D/2
1840 KE=EXP(-2*PI*AET/1.305/D*SQR(1-(1.305*D/L)^2))
1850 REM QE IS THE LARGE APERTURE CORRECTION FOR THE ELECTRIC TERM
1860 QE=PI*1.305*D/2/L

```

```
1870 QE=TAN(QE)/QE
1880 REM KM IS THE THICKNESS CORRECTION FOR THE MAGNETIC TERMS
1890 AMT=1.0064*T+.0819*D/2
1900 KM=EXP(-2*PI*AMT/1.705/D*SQR(1-(1.705*D/L)^2))
1910 REM QM IS THE LARGE APERTURE CORRECTION FOR THE MAGNETIC TERMS
1920 QM=PI*1.705*D/2/L
1930 QM=TAN(QM)/QM
1940 U1=SIN(PI*X)^2/6*KM*QM
1950 U2=COS(PI*X)^2*LG*LG/24/A/A*KM*QM
1960 U3=-SIN(PI*X)^2*LG*LG/12/L/L*KE*QE
1970 U=2*PI*D*D/A/B/LG*(U1+U2+U3)
1980 C=20*LOG(1/U)/2.302585
1990 RETURN
```

**DOCUMENT CONTROL DATA SHEET**

Security classification of this page :

**UNCLASSIFIED**

|                           |             |  |              |
|---------------------------|-------------|--|--------------|
| <b>1 DOCUMENT NUMBERS</b> |             | <b>2 SECURITY CLASSIFICATION</b>               |              |
| AR Number:                | AR-006-736  | a. Complete Document:                          | Unclassified |
| Series Number:            | ERL-0559-RN | b. Title in Isolation                          | Unclassified |
| Other Numbers:            |             | c. Summary in Isolation                        | Unclassified |
|                           |             | <b>3 DOWNGRADING / DELIMITING INSTRUCTIONS</b> |              |
|                           |             |  |              |

**4 TITLE**

THE DESIGN OF TOPWALL WAVEGUIDE DIRECTIONAL COUPLERS

**5 PERSONAL AUTHOR (S)**

Kenneth M. Harvey

**6 DOCUMENT DATE**

June 1991

**7 7.1 TOTAL NUMBER OF PAGES**

14

**7.2 NUMBER OF REFERENCES**

4

**8 8.1 CORPORATE AUTHOR (S)**

Electronics Research Laboratory

**9 REFERENCE NUMBERS**

a. Task: DST 89/116  
b. Sponsoring Agency: DST

**8.2 DOCUMENT SERIES and NUMBER**

Research Note  
0559

**10 COST CODE**

**11 IMPRINT (Publishing organisation)**

Defence Science and Technology  
Organisation Salisbury

**12 COMPUTER PROGRAM (S)  
(Title (s) and language (s))**

**13 RELEASE LIMITATIONS (of the document)**

Approved for Public Release

Security classification of this page :

**UNCLASSIFIED**

Security classification of this page :

UNCLASSIFIED

**14 ANNOUNCEMENT LIMITATIONS** (of the information on these pages)

No limitation.

**15 DESCRIPTORS**

**16 COSATI CODES**

|                           |   |
|---------------------------|---|
| a. EJC Thesaurus<br>Terms | Waveguide couplers<br>Directional couplers<br>Computer aided design |
|---------------------------|---|

|                |
|----------------|
| 090105<br>1314 |
|----------------|

|                             |
|-----------------------------|
| b. Non - Thesaurus<br>Terms |
|-----------------------------|

**17 SUMMARY OR ABSTRACT**

(if this is security classified, the announcement of this report will be similarly classified)

This report outlines a procedure for the design of topwall waveguide directional couplers. It is accompanied by a listing of a GWBASIC programme WGCUP.BAS that renders the whole procedure quick, easy and accurate.

Security classification of this page :

UNCLASSIFIED